

# REMES:EU

## User guide

Paolo Pisciella

---

# *REMES:EU – a Regional Equilibrium Model with focus on the Energy System.*

---

---

**Abstract:** The REMES model represents a multi-regional, multi-sectoral recursive dynamic Computable General Equilibrium model. It has been designed primarily to analyze the impacts of different climate policy measures and the interactions between the economies of the European Countries. The incorporation of fuel substitution allows to handle regional issues such as decentralized energy production or transmission needs or to investigate the interaction between the energy sector and the broader economy. The model is flexible in term of its input structure and can accommodate for different geographical and sectorial scopes and it allows for a flexible modelization of new sectors. It considers the effects of different GHG budget constraints and helps analysing the resulting CO2 prices. This user guide provides an overview of the CGE model, its data input structure and describes how it can be utilized for policy analysis.

---

## Contents

Section 1: Introduction.....	3
Section 2: Data .....	4
Section 2.1: Social Accounting Matrix.....	4
Section 2.2: REMES:EU input data.....	5
Section 3: CGE model and implementation .....	7
Section 3.1: Code structure .....	8
Section 3.2: Output .....	14
Section 4: Parameters and complementary information.....	15

## Section 1: Introduction

*Computable general equilibrium* models are economic models using real data to describe the joint behaviour of a group of agents and production sectors as a result of a given quantifiable policy, such as the introduction of a tax or a law limiting the usage of a given resource. This class of models utilizes data drawn from National Accounts describing supply and use of different commodities and services bundled into a so-called Social Accounting Matrix (SAM), which provides a snapshot of the monetary flows between the different actors and production sectors at the end of a given year. The behaviour of each agent and production sector is modelled based on microeconomic theory. Namely, consumers and production sectors are equipped respectively with utility functions and production functions characterized by the relative elasticities of substitution. These functions describe the behaviour of the economic agents by characterizing their propension to exchange one commodity or group of commodities with another as their relative price changes. Assuming that the economy is at an equilibrium state from which no actor has incentive to deviate unilaterally, CGE models can be used to estimate the effects of external impacts such as tax changes, investments, technological changes, policies, but also more severe shocks such as price changes of important export goods, on the economy. As the various actors respond to these changes by adapting their consumption or production (or factors of production such as labour or capital), one can investigate how these responses will propagate through the economy and to other regions in the form of shifting supply and demand for goods – until the equilibrium in the economy is re-established.

The REMES:EU model has been developed as a collaboration between SINTEF and NTNU and has been used for providing macroeconomic analyses in EU projects such as SetNav. It contains information about the internal and cross national economic flows of 29 European Countries, composed of the EU27 countries except for Croatia but with the addition of Switzerland and Norway. The model considers various real economic variables such as production, intermediate and final consumption, resources evaluation, trade, trade margins, taxes and subsidies. REMES:EU models the behaviour of both production sectors and final consumers as they tend to establish an equilibrium between demand and supply of each commodity in the economy as a consequence of a given change in the economic system conditions. These changes can be the definition of taxes or subsidies, the variation in availability of one or more resources, changes in production technology or consumption preference, as well as sectoral productivity levels. Resources explicitly modelled in REMES:EU include natural resources availability, labour force availability and carbon allowances budget. The model can be used for analysing the impact of European policies on the economy of the different countries in terms of GDP, prices, sectoral value added, unemployment, sectoral and commodity monetary input/output and CO2 emissions. The model is implemented in the General Algebraic Modelling System, or GAMS, a widely known high-level programming language and development environment (IDE). GAMS allows for the definition of mathematical models analogously as done on paper, defining sets, parameters, variables and connecting every part via algebraic equations and inequalities. The main usage of this system is in implementing mathematical programs, such as linear, non-linear and mixed integer programs. These mathematical programs are hence solved using a dedicated external solver, which is selected depending on the problem structure. The remainder presents a concise user guide for understanding

the various parts of the REMES:EU model implementation, with a description of the input data, the model itself and an explanation of the output results of the model.

## Section 2: Data

REMES:EU utilizes monetary flows data from three different sources:

- Country specific Social Accounting Matrices (SAMs)
- International trade flows data
- International trade margins flows data

These inputs define a snapshot of all the exchanges between production sectors and final consumers within and between each considered country in a base year. Namely, while the SAMs are focused on providing information of the monetary flows within each country, the data tables related to international trade and international trade margins reconcile and “glue” the different SAMs into a harmonized dataset for the whole Europe.

### Section 2.1: Social Accounting Matrix

A Social Accounting Matrix is a representation of the national accounts for a given country or region which provides a snapshot in a particular moment in time of the transactions between all the economic agents populating the considered economy. The SAM is usually represented as a square matrix whose elements can be grouped into seven categories: commodities (and services), production sectors, final consumers, resources, taxes and subsidies, international trade and finally trade margins. An element of such matrix represents the flow of money paid by the transactor represented in the column to the transactor represented in the row. Alternatively, the element in position  $(i,j)$  of the SAM represents the amount of money transferred from index  $j$  to index  $i$ .

A square SAM must satisfy the property of balancedness, i.e. the sum of the elements in a given column must be equal to the sum of elements in the corresponding row. This implies that each agent in the economy, whether a production sector or a final consumer receives money from different sources and places money to different sources (consumption of different products, transfers, taxes, savings etc...), and the sum of all the money received needs to be the same as the sum of all the money allocated to different uses. An example of structure of a SAM and its division into submatrices is reported in Figure 1.

Receipts	Expenditure				
	Activity	Commodity	Factors	Institutions	World
Activity		Domestic sales			
Commodity	Intermediate inputs			Final demand	Exports
Factors	Value added (wages/rentals)				
Institutions	Indirect taxes	Tariffs	Factor income		Capital inflow
World		Imports			
Totals	Total costs	Total absorption	Total factor income	Gross domestic income	Foreign exchange inflow

*Figure 1 General structure of a Social Accounting Matrix*

The monetary figures are read as flowing from the column to the row. As a mention, Intermediate inputs describes the flow of money that is used from activities to purchase commodities, whereas domestic sales represents the monetary flows that the sale of commodities (intended as both products and services) provide to activities.

## Section 2.2: REMES:EU input data

The data utilized to build the Social Accounting Matrix in REMES:EU is based on the 2007 version of the EXIOBASE dataset, developed under the CREEA project. The database has been defined by collecting and harmonizing national account data for a large number of European and foreign countries disaggregating the sectoral structure to increase the granularity of information. REMES draws from the Supply and Use tables featured in this database to build SAMs for 29 European countries, namely Austria, Belgium, Bulgaria, Czech Republic, Germany, Denmark, Estonia, Spain, Finland, France, Greece, Hungary, Ireland, Italy, Lithuania, Luxembourg, Latvia, Malta, The Netherlands, Poland, Portugal, Romania, Sweden, Slovenia, Slovakia, United Kingdom, Switzerland and Norway. The commodities in each country from the EXIOBASE dataset are aggregated into the following 32 macrocommodities while the initial sectors are aggregated into 24 sectors. It is possible to further aggregate the commodities and the sectors within the REMES model, depending on the case studies to consider. The input data consists of three datasets for monetary flows and a list of elasticities for substitution and transformation used to model the sectoral and consumption behaviour. The three datasets containing information on monetary flows are a Social Accounting Matrix (SAM) for each modelled country, an international trade flows matrix and a matrix of trade margins. The structure of the national SAMs in REMES:EU is as shown in Figure 2.

	Sec	Com	Lab	Cap	Tax_sec	Tax_com	Hous	Gov	Inv	Stocks	Tmarg	Trade
Sec		XDZ										
Com	IOZ						CZ	CGZ	IZ	SVZ	TMXZ	EROWZ
Lab	LZ											
Cap	KZ											
Tax_sec	TAXPZ											
Tax_com		TAXCZ										
Hous			L	K				TRANSF				TRHROW-
Gov					T_S	T_C						TRROW-
Inv							SH	SG				SROW-
Stocks									SVZ			
Tmarg		TMCZ										
Trade		MROWZ					TRHROW+	TRROW+	SROW+			

Figure 2 Structure of the SAM for each country in REMES:EU

Where we have the following definitions for the parameters

IOZ is the value of the intermediate inputs; LZ and KZ are the values of the wages paid and the profits obtained by each sector; TAXPZ is the amount of money paid as taxes by the sector; XDZ is the sectoral output in Million Euros, i.e. the flow of money that each sector obtains by the sale of the produced commodities or services; TAXCZ is the amount of taxes paid on purchase of products and services; TMCZ are the transports and trade margins paid when purchasing the commodities. These are money flows that are paid for retail, commercialization services and transport services on top of the production cost of the product or service; L and K are the total values for wages and profits in the economy; T\_S and T\_C are the total amounts paid as taxes for production and products/services respectively; CZ represents the total consumption of products and services by the households. SH represents the savings of the households; TRHROW+ is the transfers that the households send abroad; CGZ represents the public expenditure on products/services; TRANSF represents the transfers from the government to the households, SG represents the governmental savings; TRROW+ is the transfers that the government sends abroad; IZ is the consumption of products/services for investment purposes; SVZ defines the changes in stocks; SROW+ is the transfers that investors send abroad; TMXZ are the transports and trade margins received from the sale of value added services such as transport, marketing and transport services; EROWZ represent monetary income from the export of products and services; TRHROW-, TRROW- represent the net transfers to households and government coming from abroad; SROW- represents the savings from the rest of the World towards the considered economy. Besides the Social Accounting Matrices, the national accounts data is used to define two additional matrices: an international trade matrix and an international transport and trade margins (TTM) matrix. The trade matrix is composed of the elements  $TRADEZ_{g,r,\bar{r}}$  defining the monetary flows for the trade of each commodity  $g$  between country  $r$  and country  $\bar{r}$ , while the TTM matrix contains the elements  $TMCRZ_{g,r,\bar{r}}$  defining the amount paid as compensation of transport and trade margins for international transactions of commodity  $g$  between countries  $r$  and  $\bar{r}$ .

These elements need to satisfy some balancing conditions in order to be coherent with the Social Accounting Matrices of each considered country.

For each commodity and country, the sum of internal sales and the exports to all the other considered counties and the rest of the world needs to correspond to the total production of the commodity in the considered country.

For each commodity and country, the total monetary demand for the commodity needs to be equal to the trade internal to the country and the imports from the explicitly modelled countries and the rest

of the world plus the payment for the taxes and transport and trade margins both for internal production and imports.

The incoming monetary flows must be balanced with the incoming monetary flows.

The final set of inputs for the REMES:EU model is an array of elasticities used to model the behaviour of production sectors and the final consumers. Elasticities are dimensionless parameters that describe the propensity to exchange one commodity or a class of commodities with another as the relative prices change. These values are used to characterize the production function of the different sectors and the consumption (utility) functions of the final consumers. In this respect, elasticities contribute to define the technology of a sector when it comes to defining the input mix that is needed to reach a certain level of production or define the consumption basket of a consumer group. In REMES, all production and demand functions are assumed to have constant elasticities (CET, CES). These elasticities must be decided beforehand such that the resulting production and demand functions reflect reality as good as possible. Production and utility functions can also be recursively nested inside other CES/CET functions, with a different elasticity of substitution/transformation for each nest. Usually, production functions are defined by iteratively nesting production factors, starting by capital and labour. The resulting aggregate output is used as input for second nest with energy using another elasticity of substitution. This aggregate is further used as input to join it with the materials nest, with a third elasticity of substitution. This structure is widely used in CGE models and is referred to as KLEM (capital, labour, energy, materials) structure.

### Section 3: CGE model and implementation

REMES:EU is modelled as Mixed Complementarity Problem, a mathematical structure that allows to define complementary conditions linking mathematical expressions in form of inequalities to non-negative decision variables to ensure that either the inequality is satisfied as equality or the connected variable is zero. Many economic problems can be expressed as complementarity problems. In REMES:EU the complementarity structure is defined to enforce three conditions:

- Zero profit – implies that no production activity makes a positive profit.
- Market clearance – requires that supply is not smaller than demand for each commodity
- Income balance – requires that all the expenditure of the consumers equal the income

The first condition is to be understood as no activity obtaining extra-profit other than the repayment of the employed capital. This condition is linked to the property of irreversibility, i.e. all the activities are operated at non-negative levels. The second condition is associated to the existence of a non-negative price for the considered commodity. REMES mathematical formulation corresponds to an Arrow-Debreu macroeconomic model implemented using the *Mathematical Programming System for Equilibrium Analysis* (MPSGE), which is an extension to the GAMS language taking information about the diverse entities of an equilibrium model in templates, which are later “translated” into GAMS code. Regular GAMS code and the MPSGE additions are then merged and compiled by GAMS into solver-readable data to be processed.

The monetary flows featured in the model are represented in Figure 3.




Figure 3 REMES:EU main elements and monetary flows

In the diagram, traders buy from internal production, imports and transport and trade margins and sell to everyone else (consumers, government, investors and industries). Households buy from traders, send savings to investments according to a pre-specified propensity for savings and pay taxes. On the other hand, they receive money from transfers and endowments from the capital and labour market. Government buys goods and services from traders and sends transfers to households, receiving money flows from taxes. Investors buy from traders and receive savings from Government and Consumers. Consumers, government and investors behaviours are modelled according to a Cobb-Douglas utility function. Industry buys materials and energy commodities from traders, repays labour, capital and taxes, while receiving money from traders and governmental subsidies. Finally, the rest of the world buys from industries (exports) and sell to traders (imports). Capital and labour are assumed as mobile across both sectors and regions. The production of goods is represented through a three nests constant-elasticity-of-substitution (CES) function, assuming a typical KLEM structure. The model particularly emphasizes the role of natural resources such as oil, gas and coal as well as productivity from renewable sources in energy production.

### Section 3.1: Code structure

The implementation code of REMES:EU is composed of the following routines:

1. REMES\_Base.gms: it is the main routine and the GAMS program needs to be run from this file. This routine contains the data management and the definition of the case studies to be simulated. It is composed of the following blocks, which are also located in the code itself:
  - a. Choice of the user case and definition of main parameters for user case
  - b. Definition of parameters to use in the model



- c. Data reading
  - d. Reaggregation of commodities and sectors.
  - e. Assignment of parameters
  - f. Balance checks
  - g. Calculation of initial tax rates
  - h. Calculation of baseline CO2 emissions
  - i. Trade Balance checks
  - j. Read input from openENTRANCE platform
  - k. Translation of openENTRANCE input into REMES technology shares
  - l. Inclusion of the CGE model and benchmark test
  - m. Base year output definition
  - n. Recursive step definition
  - o. Application of the technology shock
  - p. Solution of the n-th iteration
  - q. Output definition
2. InputParameters.gms: the file contains the user defined parameters and reads the external dataset.
  3. growth\_table.gms: the file contains the information about baseline growth for the economy and the population.
  4. REMES\_MPSGE\_alt.gms: the file with the CGE model definition.
  5. reset\_initial\_values\_basic.gms: a file with the starting value of the variables.

Besides the REMES\_Base.gms file, the remaining ones are self explanatory and we do not provide a list of the main blocks as we have done for the main routine. We instead focus on the description of file REMES\_MPSGE\_alt.gms, containing the implementation of the model. Namely, we report the implementation of the model with a brief description of the code for each MPSGE block. Each sector is identified by a “PROD:” MPSGE block in which each input and output is assigned a price, a representative quantity and the current level of taxes applied to each input and to the output. These blocks describe the structure of the (nested) production function. Not only sectors are modelled using these structures, but also consumers preferences by means of utility functions. Consumers and other entities equipped with a budget and a utility function is also equipped with endowments and a budget constraint. These are modelled using a “DEMAND:” block. In addition, custom variables can be declared in “CONSTRAINT:” blocks to express, for example, dynamically determined taxes or scaling indices. The MPSGE code is structured as follows.

## Households

There is one household per country, representing the aggregated consumption. Households are characterized by a PROD and a DEMAND function.

The DEMAND block specifies the sources of financing for the household as endowments “E:”. Endowments include income from labour and capital, transfers from the active level of government to the households, payments from the ROW, as well as a “negative endowment” which represents the values from savings going into the investment sector. All these transactions are put together to form a budget with which the households can demand goods, “D:”. Notice that governmental transfers and savings have reference variables “R:”. These are defined in later blocks and serve as indexes for those values, so that an increase in the reference is adequately scaled in the affected entry.

```

$DEMAND:Hous (cnt)
E:PL (cnt) Q: (LSZ (cnt) * (1-ty (cnt)) *gdp_p (cnt) / (1-urate (cnt)))
E:PL (cnt) Q: (- (LSZ (cnt) * (1-ty (cnt)) *gdp_p (cnt)) / (1-urate (cnt))) R:UR (cnt)
E:RKC (cnt) Q: (KSZ0 (cnt) * (1-ty (cnt)) *gdp_p (cnt))
E:POG (cnt) $AAGR (cnt) Q: (AAGR (cnt))
E:PCL (cnt) $ACR (cnt) Q: (ACR (cnt))
E:PNG (cnt) $ANGR (cnt) Q: (ANGR (cnt))
E:PTR (cnt) Q: (TRANSFZ (cnt) * (alpha+(1-alpha)*gdp_p (cnt))) R:PCINDEX (cnt)
E:PS (cnt) Q: (-SHZ (cnt) * (alpha+(1-alpha)*gdp_p (cnt))) R:R_SH (cnt)
E:ER (CRR) $ (used_currency (cnt, CRR)) Q: (TRHROWZ (cnt) * (alpha+(1-alpha)*gdp_p (cnt)))
D:PU (cnt) Q: (CBUDZ (cnt))

```

The PROD block for the households transforms the households' budget into consumption. The inputs are the worth of the goods purchased and the mark up on these goods from transport and trade margins (TTMs), both equally adjusted for consumption tax rates. Using these means, households "consume" just enough to spend a budget, which should match what they earn in the DEMAND block. The elasticity to combine consumption and TTMs is zero, while the elasticity to combine the goods has been set to one. Taxes on the inputs are assigned to the national government, if it is active, otherwise these go to the local government

```

$PROD:U (cnt) $ (P_S > 0) s:1 com.tl(s):0
O:PU (cnt) Q: (CBUDZ (cnt) + Pathway_total (cnt, "Hous") $ (Pathway_total (cnt, "Hous") < 0))
I:P (cnt, com) $ (P_S > 0 and FF (com) = 0 and PID (com) ne 2) Q: (CZ (cnt, com) + Pathway_adj (cnt, "HOUS", com) )
+ P: (1+taxcz (cnt, com)) A:GOVTL (cnt) T:taxc (cnt, com)
I:P (cnt, com) $ (P_S > 0 and FF (com) = 0 and PID (com) eq 2)
+ Q: ((CZ (cnt, com) + Pathway_adj (cnt, "HOUS", com)) * eint (cnt))
+ P: ((1+taxcz (cnt, com)) / eint (cnt)) A:GOVTL (cnt) T:taxc (cnt, com)
I:P (cnt, com) $ (P_S > 0 and FF (com) = 1)
+ Q: ((CZ (cnt, com) + Pathway_adj (cnt, "HOUS", com)) * eint (cnt))
+ P: ((1+taxcz (cnt, com)) / eint (cnt)) A:GOVTL (cnt) T:taxc (cnt, com) com.tl:
I:PCO2 (cnt) # (com) $ (CO2H (cnt, com) and FF (com) = 1 and CO2B (cnt))
+ Q: ((CZ (cnt, com) + Pathway_adj (cnt, "HOUS", com)) * CO2r (cnt, com, "HOUS") * cint (cnt) * eint (cnt))
+ P: (1e-6 / eint (cnt)) com.tl:

```

The utility function is defined as a Cobb-Douglas with a Leontief subnest to force the complimentary purchase of  $CO_2$  allowances alongside fossil fuels. Moreover, for energy commodities, the `eint (cnt)` factor represents the learning curve in usage of energy for the considered country.

Linking the demand and the production block for the consumer we have the welfare index relationship,

$$HOUS_{cnt} = PU_{cnt} CBUDZ_{cnt} U_{cnt}$$

## Producers

Producers in the model consume intermediate goods available in the local market (marked up for taxes and the corresponding TTMs), employment of labour and capital, and natural resources. As output, they produce domestic goods. The nests of the production function follows a classic KLEM structure. Capital is aggregated with labour, then the result is aggregated with energy and finally to materials. Taxes on input goods, as well as taxes/subsidies on output goods are assigned to the governments. Similarly to the households utility function, in the energy nest, fossil fuels are aggregated with  $CO_2$  allowances using a Leontief nest and `eint (cnt)` factor represents the learning curve in usage of energy for the considered country.

```

$PROD:XD (cnt, sec) $ (XDZ (cnt, sec) and P_S > 0 and not sameas (sec, "i-H2S") and not sameas (sec, "i-H2CCS") and
not sameas (sec, "i-H2E") and not sameas (sec, "i-POW") and not sameas (sec, "i-PCCS") and not ressecs (sec))
+ t:0 s:ELAS (sec, "KLEM") sM(s):0 skle(s):ELAS (sec, "KLE") sE(sKLE):1 skl(skle):ELAS (sec, "KL") sk(skl):10
com.tl (sE):0
O:PD (cnt, sec, com) $ (XDDZ (cnt, sec, com)) Q: (XDDZ (cnt, sec, com))
+ P: (1-taxpz (cnt, sec))
+ A:GOVTL (cnt) T:taxp (cnt, sec)

```

```

* Materials and services
I:P(cnt,com)$ (PID(com) eq 0 or PID(com) eq 3)          Q:(IOZ(cnt,com,sec))
+      P:(1+taxcz(cnt,com))
+      A:GOVTL(cnt)          T:taxc(cnt,com)
+      sM:

* Non-fossil fuels
I:P(cnt,com)$ (PID(com) = 1 and (IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com)) and FF(com)=0 )
+      Q:((IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com))*eint(cnt))      P:((1+taxcz(cnt,com))/eint(cnt))
+      A:GOVTL(cnt)          T:taxc(cnt,com)
+      sE:

* Fossil fuels
I:P(cnt,com)$ (PID(com) = 1 and (IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com)) and FF(com)=1)
+      Q:((IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com))*eint(cnt))      P:((1+taxcz(cnt,com))/eint(cnt))
+      A:GOVTL(cnt)          T:taxc(cnt,com)
+      com.tl:

* CO2 allowances
I:PCO2(cnt)#(com)$ (PID(com) = 1 and CO2P(cnt,com,sec) and FF(com)=1 and CO2B(cnt))
+      Q:((IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com))*CO2r(cnt,com,sec)*cint(cnt)*eint(cnt))      P:(1e-
6/eint(cnt))
+      com.tl:

* Hydrogen
I:P(cnt,"gH2")$(Pathway_adj(cnt, sec, "gH2"))
+      Q:(Pathway_adj(cnt, sec, "gH2")*eint(cnt))      P:(1/eint(cnt))
+      sE:

* energy generation
I:P(cnt,com)$ (PID(com) = 2 and (IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com)))
+      Q:((IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com))*eint(cnt))      P:((1+taxcz(cnt,com))/eint(cnt))
+      A:GOVTL(cnt)          T:taxc(cnt,com)
+      sE:

* energy transmission
I:P(cnt,com)$ (PID(com)=4 and (IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com)))
+      Q:((IOZ(cnt,com,sec)+Pathway_adj(cnt, sec, com))*eint(cnt))      P:((1+taxcz(cnt,com))/eint(cnt))
+      A:GOVTL(cnt)          T:taxc(cnt,com)
+      sE:

* Labour, Capital & Savings
I:PL(cnt)          Q:(LZ(cnt,sec)/PRC(cnt))      skl:
I:RKC(cnt)$ (Pathway_adj_cap(cnt, sec))          Q:(Pathway_adj_cap(cnt, sec)/PRC(cnt))      sk:
I:PS(cnt)          Q:(INVZ(cnt,sec)      sk:

```

The production blocks slightly change for natural resources. Namely, at the top of the nesting structure, there is a further aggregation with the input of the natural resource to be extracted. This is useful to simulate a cap on the availability of the natural resource, for example as a consequence of a law limiting the extractions. The domestic production obtained as output of the production sectors is fed into an other function splitting this production into the different geographical markets it is targeting. Namely, a function for the definition of the exports accepts as input the production of a given commodity from different sectors according to a CES function and produces as output the amount of the aforementioned commodity allocated into the internal market, the explicitly modelled EU countries and the rest of the world. In the following block the input variable  $XDDZ(cnt, sec, com)$  for commodity com and country cnt, produced from different sectors is either allocated in the internal market ( $XXDZ(cnt, com)$ ) sold to other modelled EU countries ( $TRADEZ(com, cnt, cntt)$ ) or sold to the rest of the World ( $EROWZ(cnt, com)$ ).

```

$PROD:EXPORT(cnt,com)$ ( sum(sec, XDDZ(cnt,sec,com) ) and not sameas(com,"gH2") and not sameas(com,"gPOW"))
t:1.4$(worldcom(com)=0) t:5$(worldcom(com)=1) s:1.2
O:ER(CRR)#(cntt)$ (used_currency(cnt, CRR))          Q:(TRADEZ(com,cnt,cntt))
O:PDD(cnt,com)$ (XXDZ(cnt,com)>1e-6)          Q:XXDZ(cnt,com)
O:ER(CRR)$ (used_currency(cnt, CRR))          Q:EROWZ(cnt,com)

```

I:PD(cnt, sec, com)\$(XDDZ(cnt, sec, com))

Q:(XDDZ(cnt, sec, com)\*(t\*\*xdsp))

In the block, ER defines the exchange rate and is related to a given currency CRR. Parameter xdsp is used to gradually make the export sector more inefficient and lead the model to another export sector, normally exporting the same commodity with a different input structure. This is typical for sectors that are not active in the base year, but will become active in the counterfactual analyses.

## Goods

This block, for each country and commodity, aggregates imports and domestic production into a representative good that is consumed in the considered country. This is done in accordance to the so-called Armington assumption. For goods produced locally, or in other modelled countries, the block aggregates the value of the trade margin paid to the output price of the Armington good. The block is as follows

```
$PROD:X(cnt,com)$(XZ(cnt,com) and not sameas(com,"gH2")) s:elasM(com) s2:0 s3:0
O:P(cnt,com) Q:(XZ(cnt,com))
I:PDD(cnt,com)$(XXDZ(cnt,com)>1e-6) Q:XXDZ(cnt,com) s2:
I:PTM(cnt,cntt)$(ord(cnt)eq ord(cntt)) Q:(trademargins(com,cntt,cnt)) s2:
I:ER(CRR)$(used_currency(cnt,CRR)) Q:MROWZ(cnt,com)
I:ER(CRR)#(cntt)$(used_currency(cnt,CRR)) Q:TRADEZ(com,cntt,cnt) A:GOVTL(cnt) T:tfp(cntt) s3:
I:PTM(cnt,cntt)$(ord(cnt)ne ord(cntt)) Q:trademargins(com,cntt,cnt) s3:
```

Trade margins are defined in a transport block, which aggregates a number of services and goods to produce trade and transport services as follows

```
$PROD:TRANSP(cnt) s:0 t:0
O:PTM(cnt,cntt)#(com) Q:(trademargins(com,cntt,cnt))
I:P(cnt,com) Q:(TMXZ(cnt,com))
```

## Government

The government, like the households, has both a producer and a demand block. The government's demand block is characterized by a) a series of endowments, and b) taxes received from all entities; these go to form its budget. The government has income from taxes, CO2 allowances and transfers from the rest of the World. Income from taxes is not explicitly modelled in the block but it is automatically taken into account by MPSGE. The government's endowments are the transfers from the rest of the world

```
$DEMAND:GOVTL(cnt)
E:PCO2(cnt)$(CO2B(cnt)) Q:CO2B(cnt)
E:RKC(cnt)$(sum(sec,pathway_adj_cap(cnt,sec)) and (sum(sec,pathway_total(cnt,sec))<0 and P_S>0))
+ Q:(-sum(sec,pathway_total(cnt,sec))*(alpha+(1-alpha)*gdp_p(cnt)))
E:PS(cnt) Q:(-SGZ(cnt)*(alpha+(1-alpha)*gdp_p(cnt))) R:PIINDEX(cnt)
E:ER(CRR)$(used_currency(cnt,CRR)) Q:(TRROWZ(cnt)*(alpha+(1-alpha)*gdp_p(cnt)))
E:PTR(cnt) Q:(-TRANSFZ(cnt)*(alpha+(1-alpha)*gdp_p(cnt))) R:PCINDEX(cnt)
D:PUGL(cnt) Q:(CBUDGLZ(cnt)-sum(sec,pathway_total(cnt,sec)))
```

The production function or utility of the government actor(s) is similar to that of the households. The government consumes goods sold in the country in order to produce a budget which should match that in the demand block.

```
$PROD:UGL(cnt)$(P_S > 0) s:1 com.tl(s):0
O:PUGL(cnt) Q:(CBUDGLZ(cnt) + Pathway_total(cnt, "Govt")$(Pathway_total(cnt, "Govt")<0))
I:P(cnt,com)$(P_S > 0 and FF(com) = 0 and PID(com) ne 2)
```

```

+ Q: (CGLZ(cnt,com) + Pathway_adj(cnt, "GOVT", com)) P: (1+taxcz(cnt,com)) A:GOVTL(cnt) T:taxc(cnt,com)
I: P(cnt,com)$(P_S > 0 and FF(com) = 0 and PID(com) eq 2)
+ Q: ((CGLZ(cnt,com) + Pathway_adj(cnt, "GOVT", com))*eint(cnt)) P: ((1+taxcz(cnt,com))/eint(cnt))
+ A:GOVTL(cnt) T:taxc(cnt,com)
I: P(cnt,com)$(P_S > 0 and FF(com)=1)
+ Q: ((CGLZ(cnt,com) + Pathway_adj(cnt, "GOVT", com))*eint(cnt)) P: ((1+taxcz(cnt,com))/eint(cnt))
+ A:GOVTL(cnt) T:taxc(cnt,com) com.tl:
I: PCO2(cnt)#(com)$(CO2G(cnt,com) and FF(com)=1 and CO2B(cnt))
+ Q: ((CGLZ(cnt,com) + Pathway_adj(cnt, "GOVT", com))*CO2r(cnt,com,"GOVT")*cint(cnt)*eint(cnt))
+ P: (1e-6/eint(cnt)) com.tl:

```

Similarly as for Households we can link the demand and the production block with the relationship

$$GOVTL_{cnt} = PUGL_{cnt}CBUDGLZ_{cnt}UGL_{cnt}$$

## Investment

The investment sector turns the purchase of different goods and services into new capital. Just as for households and government is defined as both a production and a demand block. The demand block takes as endowments the savings of the households and government acting in the country, properly scaled, as well as the investment sent from abroad and the enterprise investments. Exogenous changes in stocks act as negative endowments to this block, too. All this forms the investment budget, which will be distributed in the production block of the investment sector. For investment, the demand block is as follows

```

$DEMAND: INVB(cnt)
E: PS(cnt) Q: (SHZ(cnt)*(alpha+(1-alpha)*gdp_p(cnt))) R: R_SH(cnt)
E: PS(cnt) Q: (SGZ(cnt)*(alpha+(1-alpha)*gdp_p(cnt))) R: PIINDEX(cnt)
E: ER(CRR)$(used_currency(cnt, CRR)) Q: (SROWZ(cnt)*(alpha+(1-alpha)*gdp_p(cnt)))
E: P(cnt,com) Q: (-SVZ(cnt,com)*(alpha+(1-alpha)*gdp_p(cnt))) R: R_SV(cnt,com)
E: PS(cnt) Q: (sum(sec, INVZ(cnt,sec)*XD.L(cnt,sec))*gdp_p(cnt))
D: PUINVB(cnt) Q: (ITZ(cnt))

```

While the production block, representing the preferences we have

```

$PROD: UINVB(cnt) s:1 com.tl(s):0
O: PUINVB(cnt) Q: ITZ(cnt)
I: P(cnt,com)$(FF(com)=0 and PID(com) ne 2) Q: (IZ(cnt,com)) P: (1+taxcz(cnt,com))
+ A:GOVTL(cnt) T:taxc(cnt,com)
I: P(cnt,com)$(FF(com)=0 and PID(com) eq 2) Q: (IZ(cnt,com)) P: (1+taxcz(cnt,com))
+ A:GOVTL(cnt) T:taxc(cnt,com)
I: P(cnt,com)$(FF(com)=1) Q: (IZ(cnt,com))
+ P: (1+taxcz(cnt,com)) A:GOVTL(cnt) T:taxc(cnt,com) com.tl:
I: PCO2(cnt)#(com)$(CO2I(cnt,com) and FF(com)=1 and CO2B(cnt))
+ Q: ((IZ(cnt,com))*CO2r(cnt,com,"INV")*cint(cnt)) P: 1e-6 com.tl:

```

Linked by

$$INVB_{cnt} = PUINVB_{cnt}ITZ_{cnt}UINVB_{cnt}$$

The investment level in one run of the model is converted into capital and increases the amount of pre-existing capital after depreciation, according to the formulas

```

Kap(cnt) = (Kap(cnt)*(1-depr(cnt))+II.l(cnt)*5);
KSZ(cnt) = Kap(cnt)*ror(cnt);

```

Where  $II.l(cnt)$  is the value of investments after one iteration of the model is made. This value is multiplied by 5 because we consider 5 years steps, therefore we simulate a linear increase over the

next 5 years. `ror(cnt)` and `depr(cnt)` are respectively the rate of return of capital and the depreciation rate for a given country.

## Goods

All the fields tagged with an **R**: in the demand sections are used to define scaling factors applied to the relative endowment. This is meant to scale the the endowments proportionally to changes in the economy. The scaling factors are defined by the following constraints.

The first constraint defines the consumer price index, used to control the transfers from the government to the households, while the second constraint defines the scaling factor for the savings level of the government. The third and fourth constraint define the scaling factors for the changes in stocks and the savings level of the households. The last constraint defines the unemployment rate, linked to the ratio between the wages and the consumption price index. The *bin* parameter is binary and is used to activate and deactivate the constraint.

```
$CONSTRAINT:PCINDEX(cnt)
    PCINDEX(cnt) =e= sum(com, (1+taxc(cnt,com))*P(cnt,com) *CZ(cnt,com) ) /
        sum(com, (1+taxcz(cnt,com))*CZ(cnt,com) ) ;

$CONSTRAINT:PIINDEX(cnt)
    PIINDEX(cnt) =e= sum(com, (1+taxc(cnt,com))*P(cnt,com) *IZ(cnt,com) ) /
        sum(com, (1+taxcz(cnt,com))*IZ(cnt,com) ) ;

$CONSTRAINT:R_SV(cnt,com)
    R_SV(cnt,com) =e= X(cnt,com) ;

$CONSTRAINT:R_SH(cnt)
    R_SH(cnt) =e= HOUS(cnt) / (CBUDZ(cnt));

$CONSTRAINT:UR(cnt)
    PL(cnt)=g=1-bin+pu(cnt)*bin;
```

## Section 3.2: Output

The model is run from routine “REMES\_Base.gms” by selecting the case study at the top of the routine. Each case study has a set of parameters explained in the code that can be modified. The output will be a file called “outputREMES.gdx”, containing projections about GDP, sectoral Value Added, Unemployment rate, Monetary flows analysis per sector and per commodity, Social Accounting Matrices, relative price levels for commodities, change in supply and demand of commodities and the relative price of CO2.

Moreover a file “REMESOE2.csv” contains the data in IAMC format to be used in the openENTRANCE scenario platform. Once the model simulation is terminated, the results are converted into the IAMC format using the following code snippet in GAMS, which creates a csv file that is then uploaded into the Platform.


```
File remesoe2 /REMESOE2.csv;
remesoe2.tf=0;
put remesoe2;
put 'model,scenario,region,variable,unit,subannual,2010,2015,2020,2025,2030,2035,2040,2045,2050', put /;
loop((cnt)$ (not (sameas(cnt,'CZ') or sameas(cnt,'GB') or sameas(cnt,'NL'))), put 'REMES:EU 1.0,Green Growth,', put country(cnt), put '
loop(step$(ord(step) gt 1), put ',', put EneDemM(cnt,step)) put /;);
loop((cnt)$ (sameas(cnt,'CZ'))), put 'REMES:EU 1.0,Green Growth,', put 'Czech Republic', put ',Consumption,billion US$2010/yr,Year',
loop(step$(ord(step) gt 1), put ',', put EneDemM(cnt,step)) put /;);
loop((cnt)$ (sameas(cnt,'GB'))), put 'REMES:EU 1.0,Green Growth,', put 'United Kingdom', put ',Consumption,billion US$2010/yr,Year',
loop(step$(ord(step) gt 1), put ',', put EneDemM(cnt,step)) put /;);
loop((cnt)$ (sameas(cnt,'NL'))), put 'REMES:EU 1.0,Green Growth,', put 'The Netherlands', put ',Consumption,billion US$2010/yr,Year',
loop(step$(ord(step) gt 1), put ',', put EneDemM(cnt,step)) put /;);
```

This creates a csv file in the IAMC format. The output provided by GAMS features blank spaces before and after every string, which makes the output not suitable for the Scenario Platform. To solve this issue, a dedicated VBA routine has been developed in an Excel spreadsheet. The output is now accepted by the Scenario Platform. After following the instructions provided in the Guideline to Use the OpenENTRANCE database, the data file is uploaded into the Scenario Explorer. The following snapshot is taken directly on the data loaded on the Scenario Explorer.

Source						Years									
Model	Scenario	Region	Variable	Subannual	Unit	2010	2015	2020	2025	2030	2035	2040	2045	2050	
REMES-EU 1.0	Directed Transition	Europe	Consumption	Year	billion US\$2010/yr	422	461.58	509.02	535.67	583.67	639.52	702.12	772.02	852.35	
REMES-EU 1.0	Directed Transition	Europe	GDP/PPP	Year	billion US\$2010/yr	16338...	17414...	18566...	19951...	21312...	22762...	24333...	26004...	27784...	
REMES-EU 1.0	Directed Transition	Europe	Price/Final Energy/Residential/Electricity	Year	US\$2010/GJ	63.3	62.79	62.21	61.19	60.59	59.92	59.13	58.44	57.87	
REMES-EU 1.0	Directed Transition	Europe	Final Energy/Electricity	Year	EJ/yr	76186...	84142...	93848...	10021...	11038...	12248...	13647...	15210...	17014...	
REMES-EU 1.0	Directed Transition	Europe	Price/Carbon	Year	US\$2010/t CO2	4.99	18.85	39.9	25.27	36.91	49.41	63.1	77.96	93.8	

## Section 4: Parameters and complementary information

### 1. Temporal and Geographical scope



## 2. Sets in REMES

### 2.1. Regions

AT	BE	BG	CY	CZ	DE	DK	EE	ES	FI	FR	GR
HU	IE	IT	LT	LU	LV	MT	NL	PL	PT	RO	SE
SI	SK	GB	CH	NO							

### 2.2. Sectors

Agriculture, forestry and fishing	Constructions
Mining of Coal	Administration and Consultancy
Oil extraction	Retail
Mining of minerals, metals and stones	Railway Transport
Rest of industry	Land Transport
Aluminium production	Transport via pipelines
Production of electricity by thermal sources	Inland water Transport
Production of electricity by renewable sources	Sea Transport
Transmission of electricity	Air Transport
Manufacture and distribution of gas	Health sector
Steam and hot water supply	Waste treatment
Other services	Biogasification

### 2.3. Commodities

Agriculture products	Health services
Minerals	Waste treatment
Rest of production	Coal
Natural gas	Crude oil



Biofuels	Natural Gas
Electricity from renewables	Gasoline
Electricity from thermals	Jet fuel
Power transmission	Kerosene
Steam and hot water	Diesel
Other services	Heavy distillate
Constructions	Light distillate
Retail	Minerals, metals and stones
Administration and Consultancy	Sea transport
Railway Transport	Inland water transport
Land Transport	Air Transport
Pipelines	Oil

3. *List of parameters used by REMES EU (in red the parameters to obtain from energy system models)*

XDZ(cnt,sec) aggregated sectoral outputs per country

XDDZ(cnt,sec,com) detailed sectoral outputs per country

**IOZ(cnt,com,sec) intermediate inputs per sector and country**

LZ(cnt,sec) labour inputs per sector and country

KZ(cnt,sec) capital inputs per sector and country

CZ(cnt,com) households consumption composition per country

CGZ(cnt,com) governmental consumption composition per country

CGLZ(cnt, com) local government consumption composition per country

IZ(cnt,com) purchases for investments per country

SVZ(cnt,com) changes in stocks of different products per country

EROWZ(cnt,com) exports per country and commodity

MROWZ(cnt,com) imports per country and commodity

TRADEZ(com,cnt,cntt) trade flows per commodity between countries

TMCRZ(com,cnt,cntt) trade and transport margins per commodity between countries  
 TMCZ(cnt,com) transport and trade margins purchased per country  
 TMXZ(cnt,com) production of transport and trade margins per country  
 TTYZ(cnt) income taxes per country  
 TRANSFZ(cnt) government to households transfers per country  
 TRANSFGZ(cnt) national to local government transfers per country  
 SHZ(cnt) households savings per country  
 SGZ(cnt) local governmental savings per country  
 CBUDGLZ(cnt) local governmental consumption budget per country  
 SROWZ(cnt) saving transfers from RoW per country  
 INVZ(cnt,sec) sectoral investments per country  
 TRROWZ(cnt) net transfers to government (closing trade balance)  
 TRHROWZ(cnt) net transfers to households (closing trade balance)  
 TAXCZ(cnt,com) net taxes on products per country  
 TAXPZ(cnt,sec) net taxes on production per country  
 LSZ(cnt) initial labour endowment per country  
 KSZ(cnt) initial capital endowment per country  
 CBUDZ(cnt) households consumption budget per country  
 CBUDGZ(cnt) governmental consumption budget per country  
 XZ(cnt,com) total sales per commodity and country  
 XXDZ(cnt,com) domestic products supply to domestic market per commodity and country  
 ITZ(cnt) total investments per country

4. *List of variables computed by REMES EU (in red the parameters to provide to other models)*

U(cnt) Consumption of households  
 UGL(cnt) Consumption of the local governments  
 UINVB(cnt) Fixed capital investments per country  
 X(cnt,com) Commodity consumption per country  
 TRANSP(cnt) Transport services per country  
 EXPORT(cnt,com) Export activity per country and commodity  
 XD(cnt,sec) Domestic sectoral production per country

PU(cnt)	Private consumption price index per country
PUGL(cnt)	Local Governmental consumption price index per country
PUINVB(cnt)	Fixed capital local investment price index per country
PD(cnt,sec,com)	Domestic output price per commodity, sector and country
PDD(cnt,com)	Price of domestic goods provided to domestic market per country
P(cnt,com)	Composite consumer price per country and commodity
PL(cnt)	Wages per country
RKC(cnt)	Return on capital per country
ER(CRR)	Terms of trade per currency
PTM(cnt, cntt)	Transport and trade margins per country
PTR(cnt)	Price of governmental transfers per country
PS(cnt)	Price of savings per country
HOUS(cnt)	Representative household income per country
GOVTL(cnt)	Local Governments income per country
INVB(cnt)	Local Investment agents per country
PCINDEX(cnt)	Consumer price index per country
PIINDEX(cnt)	Local Investment index per country
R_SV(cnt,com)	Multiplier for changes in stocks per country
R_SH(cnt)	Multiplier for households savings per country